
CORSO DI CALCOLATORI ELETTRONICI

Informatici

7. Somma elementi pari della diagonale principale di una matrice
8. Calcolo quoziente e resto di una divisione intera

9. Calcolo della media aritmetica di 3 valori

10. Programma che imposta la parità su una sequenza di byte

11. Caricamento di un vettore con un dato carattere

12. Calcolo della somma di due vettori

13. Costruzione vettore somma delle colonne di una matrice MxN

14. Ricerca sequenziale di un elemento in un vettore

15. Ricerca binaria di un elemento in un vettore

16. Ricerca di un elemento in un vettore pari alla somma dei rimanenti

17. Calcolo approssimato della radice quadrata di un numero N

18. Dato un vettore di N (>=1 elementi) effettuare una rotazione di una posizione

dei suoi elementi
19. Calcolo della media degli elementi di due vettori di dimensione N (>=1)
20. Calcolo del massimo, e della sua posizione, tra i valori (long) della diagonale principale di una matrice quadrata di ordine N, memorizzata per righe
21. Calcolo del minimo, e della sua posizione, tra i valori (long) della diagonale secondaria di una matrice quadrata di ordine N, memorizzata per righe

22. Date due stringhe S1 e S2 effettuarne un confronto e produrre 0 se sono diverse, 1 se sono uguali

23. Data una matrice MAT RxN (dove N e R sono due costanti) di longword memorizzata per colonne a partire dalla locazione 400h, ed un vettore VETT di R elementi longword, memorizzato a partire dalla locazione 500h, calcolare il prodotto MATT x VETT e porlo  nel vettore VETT2 in 600h. La matrice e il vettore possono assumere anche valori negativi. Ignorare eventuali casi di overflow.

24. Programma di ordinamento di un vettore V di N elementi long mediante la tecnica del bubblesort

25. Confrontare due stringhe S1 e S2 restituendo il valore 1 se S1>S2, 0 se S1=S2 e -1 se S1<S2.

26. Data una matrice Mat NxM, memorizzata per colonne, stabilire se esiste una riga tutta formata da zeri.

27. Somma ricorsiva di due valori A e B
28. Data una matrice MAT  NxM, di valori interi compresi tra 0 e MAX-1, memorizzata per colonne, costruire un vettore F di MAX componenti, in modo tale che in ogni componente sia riportata la frequenza con cui il valore dell’indice di F si presenta nella matrice. 
29. Calcolo del fattoriale di N in modo ricorsivo

N.B. – Trattandosi di una prima stesura, ringrazio in anticipo coloro che volessero mandarmi note e osservazioni riguardo a sviste, imprecisioni o miglioramenti che si ritenessero necessari.

Esercizio 1
Scrivere un programma che data una stringa di caratteri alfabetici maiuscoli, conti il numero di vocali presenti

Soluzione

Analisi

Supponiamo che la stringa abbia lunghezza LUNG =4, che sia formata da caratteri alfabetici maiuscoli e che sia memorizzata nella variabile STRINGA a partire dalla locazione 400h.
Registri usati:

R0: risultato

R1: offset per l’accesso ai caratteri successivi di STRINGA

R2: singolo carattere di STRINGA da esaminare

R3: lunghezza di STRINGA da usare per la condizione di uscita

Algoritmo

Inizio
   Caricamento dati

   Azzera registri

   Inizializza R3 con LUNG

   Fintantoché (R3<>0)

   Inizio

      R2=carattere successivo di STRINGA

      Se (R2=’A’) allora R0=R0+1;   

      Se (R2=’E’) allora R0=R0+1;   

      Se (R2=’I’)   allora R0=R0+1;   

      Se (R2=’O’) allora R0=R0+1;   

      Se (R2=’U’) allora R0=R0+1;   

      R1=R1+1;

      R3=R3-1;

   Fine;
Fine.
Codifica
;Programma per il conteggio delle vocali in una stringa

;maiuscola

;Autore: M.M.

;Data: 12/02/2004

;File: contavoc.asm

ORG 400h


STRINGA DB  41h, 45h, 42h, 43h  ;stringa 'AEBC'


LUNG    DB  4           ;lungh. stringa


NUM     DL  0           ;risultato

CODE 

;****** azzeramento registri ************************



XORL R2, R2   ;azzera R2; R2 è usato 



      ;per contenere il singolo carattere


XORL R1, R1   ;offset di STRINGA


XORL R0, R0   ;azzera R0; R0 è usato per contenere 



      ;il risultato


XORL R3, R3   ;R3 conterrà la lunghezza di STRINGA

;****** inizializzazioni ****************************


MOVB LUNG, R3


CMPB #0, R3             


JZ FINE

;fine stringa

NEXT:
MOVB STRINGA(R1),R2  ;legge carattere


CMPB #41h, R2


JZ CONTA        ;se = 'A'


CMPB #45h, R2

  
JZ CONTA        ;se = 'E'


CMPB #49h, R2


JZ CONTA        ;se = 'I'


CMPB #4Fh, R2


JZ CONTA        ;se = 'O'


CMPB #55h, R2


JZ CONTA        ;se = 'U'


JMP NOVOC

CONTA:  ADDB #1, R0

NOVOC:
ADDB #1, R1


SUBB #1, R3


JNZ NEXT

FINE:
HALT


END
Esercizio 2
Scrivere un programma che dato un intero N (>=0) calcoli il valore dell’N-esimo numero della successione di Fibonacci.

Soluzione

Analisi

La successione dei numeri di Fibonacci è definita come:

F0=0, F1=1

Fi =Fi-1 + Fi-2
Registri usati:
R0: contatore dei cicli

R1: penultimo valore
R2: ultimo valore

R3: valore attuale

R4: registro di appoggio

Algoritmo

Inizio

   Azzera registri

   Inizializza registri;      ;R0=N, R1=0, R2=1

   Fintantoché (R0>=1)

   Inizio

      attuale=ultimo+penultimo;  

      penultimo=ultimo;

      ultimo=attuale;

      R0=R0-1;
   Fine;

Fine.

Codifica

;Programma che dato il numero intero N calcola il valore

;dell'N-esimo numeri della successione di Fibonacci

;La successione di Fibonacci è definita come:

;                  F0=0, F1=1

;                  F =F   + F

;                   i  i-1   i-2

;ed i primi termini sono: F0=0, F1=1, F2=1, F3=2, 

;F4=3, F5=5, F6=8 ...e così via

;Si usa Ro come contatore dei cicli e R1, R2 ed R3 come

;valori temporanei. Il ciclo termina quando R0 è arrivato

;a 0. Si supponga N>=2

;Autore: M.M.

;Data: 04/03/2004

;File: fibonac.asm

ORG 400h


N DB 7


FIBO EQU 500h

CODE

;************ azzeramento registri ******************


XORL R0, R0    ;


XORL R1, R1    ;


XORL R2, R2    ;


XORL R3, R3    ;


XORL R4, R4    ;temporaneo 

;*********** inizializzazione registri *************


MOVB N, R0


MOVB #0, R1       ;primo valore


MOVB #1, R2       ;secondo valore    


CMPB #3, R0


JN fine

ciclo:
MOVB #0, R4       ;accumulatore  


ADDB R1, R4      


ADDB R2, R4       ;attuale=ultimo +penultimo


MOVB R4, R3


MOVB R2, R1       ;penultimo=ultimo


MOVB R3, R2       ;ultimo=successivo


SUBB #1, R0


CMPB #1, R0


JNZ ciclo

fine:


HALT

END

Esercizio 3
Scrivere un programma che dati due vettori V1 e V2 di N componenti ciascuno (N>=1) ne calcoli il prodotto scalare.

Soluzione
Analisi

Dati due vettori V1 e V2 ad esempio di 3 componenti

V1=[1,2,3]

V2=[1,0,1]

Il prodotto scalare è dato dalla somma dei prodotti delle componenti omonime. 

Pertanto:

ProdScal = 1*1 + 2*0 + 3*1 = 4

Usiamo due vettori V1 e V2 di byte memorizzati rispettivamente a partire dalle locazioni 400h e 500h.

Registi usati:

R0: lunghezza dei vettori  (N)

R1: indice dei vettori (offset)

R3: elemento di V1

R4: elemento di V2

R5: temporaneo per il prodotto

R6: accumulatore per risultato

Algoritmo
Inizio

   Caricamento dati;

   Azzeramento registri;

   Inizializzazione registri;     ;R0=N

   Per R0=N  a 0 fai

   Inizio

      R3=V1[R1];

      R4=V2[R1];

      R5=R3*R4;
      R6=R6+R5

   Fine;

Fine

Codifica

;Programma che dati due vettori di N componenti,

;ne calcola il prodotto scalare

;Si esaminano una ad una, le singole componenti e si

;accumula il loro prodotto in un registro

;Autore: M.M.

;Data: 04/03/2004

;File: prodscal.asm

ORG 800h


V1 EQU 400h


V2 EQU 500h


N  EQU 3

CODE

;*********** caricamento dati **********************


MOVB #1, V1


MOVB #2, V1+1


MOVB #3, V1+2


MOVB #1, V2


MOVB #2, V2+1


MOVB #3, V2+2

;************ azzeramento registri ******************


XORL R0, R0    ;lunghezza vettori


XORL R1, R1    ;offset vettore


XORL R3, R3    ;elemento di V1


XORL R4, R4    ;elemento di V2


XORL R5, R5    ;temporaneo per prodotto


XORL R6, R6    ;risultato

;*********** inizializzazione registri *************


MOVL #N, R0

ciclo:
MOVB V1(R1), R3   ;esamina valore successivo

        MOVB V2(R1), R4   ;esamina valore successivo


MOVB #0, R5

molt:   ADDB R4, R5       ;prodotto coppia di valori

        SUBB #1, R3


JNZ molt


ADDB R5, R6       ;accumulo nel risultato


ADDL #1, R1       ;incrementa offset


SUBB #1, R0       ;decrementa contatore


JNZ ciclo


HALT

END

Esercizio 4
Data una sequenza di 8 cifre binarie convertirla in decimale.

Soluzione

Analisi

Si memorizzano in un vettore CIFRE i valori binari a partire dalla posizione più significativa posta nella locazione 400h.

Si esaminano mediante un ciclo, i singoli bit e per ciascuno si determina se è 0 oppure 1 mediante l’operazione di AND. Se si tratta di 1 il risultato verrà moltiplicato per 2 (shift a sinistra) e quindi aggiunto un 1, altrimenti si esegue soltanto la moltiplicazione per 2.

Nel risultato, al termine sarà presente il valore decimale.

Registri usati:
R0: contatore del numero di cifre per la terminazione del programma

R1: offset del vettore

R2: valore decimale (accumulatore)

R3: singola cifra binaria

Algoritmo

Inizio

   Caricamento valore binario;
   Azzeramento registri

   Inizializzazione;   R0=N
   Fintantoché (R0<>0)

   Inizio

      R3=CIFRE[R1];

      R2=R2*2;

      Se (R3=1)

         R2=R2+1; 

      R1=R1+1;

      R0=R0-1;

   Fine;

Fine.

Codifica

;Programma che data una sequenza di 8 cifre binarie 

;memorizzata in un vettore, a partire dalla cifra più

;significativa, la converta in decimale

;Si esaminano uno ad uno, i singoli bit e per ciascuno si

;esegue l'AND con il valore 1 per vedere se si tratti di 

;uno 0 o di un 1. Si aggiungono in R2 i soli bit 1 e prima

;di eseguire l'aggiunta si moltiplica R2 per 2 mediante uno

;shift a sinistra

;Autore: M.M.

;Data: 04/03/2004

;File: bintodec.asm

ORG 800h


CIFRE EQU 400h


NCIFRE EQU 8

CODE

;*********** caricamento dati **********************


MOVB #0, CIFRE


MOVB #0, CIFRE+1


MOVB #0, CIFRE+2


MOVB #0, CIFRE+3


MOVB #1, CIFRE+4


MOVB #0, CIFRE+5


MOVB #0, CIFRE+6


MOVB #0, CIFRE+7

;************ azzeramento registri ******************


XORL R0, R0    ;contatore cifre


XORL R1, R1    ;offset vettore


XORL R2, R2    ;valore decimale


XORL R3, R3    ;singola cifra binaria

;*********** inizializzazione registri *************


MOVL #NCIFRE, R0

ciclo:
MOVB CIFRE(R1), R3   ;esamina cifra successiva


ANDB #1, R3          ;test 


JNZ nonzero          ;se si tratta di un 1


ASLL #1, R2          ;moltiplica per 2


JMP continua

nonzero:


ASLL #1, R2          ;moltiplica per 2


ADDL #1, R2          ;aggiunge 1

continua:


ADDB #1, R1          ;avanza dell'offset


SUBB #1, R0          ;decrementa contatore


JNZ ciclo


HALT

END

Esercizio 5
Dato un numero intero N, calcolare la somma delle sue cifre

Soluzione

Analisi


Il numero N è rappresentato in esadecimale. Per calcolare la somma delle cifre di N si deve estrarre la cifra meno significativa, sommarla in un accumulatore e ripetere l’operazione sulla parte intera. La cifra meno significativa si ottiene eseguendo la sottrazione tra il numero e la parte intera (ottenuta mediante shift destro di 4 posizioni) moltiplicata per 16 (shift sinistro di 4 posizioni).  Il procedimento si ripete fino a quando si ottiene il valore 0 per N.
Esempio:

N=52 
Parte intera ( shift  destro di 4 posizioni ( 5
Cifra meno significativa ( 52- Parte intera *16 = 2
Registri usati: 

R0: contiene il numero N che deve essere elaborato ad ogni ciclo

R1: contiene il numero su cui operare le divisioni ad ogni ciclo

R2: accumulatore per la somma delle cifre

R3: contiene, in ogni ciclo la parte intera del numero

Algoritmo

Inizio


Azzeramento registri;


Inizializzazione registri;


R0=N;


Ripeti


    R1=R0;


    R0=R0 div 16;


    R3=R0;

    R0=R0*16;


    R1=R1-R0;

    R2=R2+R1;
 
    R3=R0;

            Fino a che (R0=0);
Fine.

Codifica

;Programma per il calcolo della somma delle cifre di un

;numero intero N

;Autore: M.M.

;Data: 15/01/2004

;File: sommacif.asm

ORG 400h


N DL 45h      ;valore decimale o esadecimale 

CODE

;********  azzeramento registri *************


XORL R0, R0


XORL R1, R1


XORL R2, R2


XORL R3, R3

;********  inizializzazione *****************

   
MOVL #0, R2


MOVL N, R0

LOOP:
MOVL R0, R1   ;prossimo valore


ASRL #4, R0   ;divide R0 per 16


MOVL R0, R3   ;salva il quoziente

 
ASLL #4, R0   ;moltiplica per 16


SUBL R0, R1   ;calcola resto (cifra)


ADDL R1, R2   ;accumula la cifra in R2


MOVL R3, R0   ;carica nuovo quoziente


CMPL #0, R0   ;se R0<>0 prosegue


JNZ  LOOP


HALT

END
    

Esercizio 6
Date due date D1 e D2 nel formato giorno/mese/anno eseguire un confronto tra esse per stabilire se D1<D2, oppure D1=D2 oppure D1>D2.

Soluzione

Analisi

Dette D1 e D2 le due date, supponiamo di indicare con G1, M1, A1 e G2, M2 e A2 rispettivamente il giorno, mese anno di esse. 

Definiamo G1, M1, G2, M2 come byte e A1 e A2 come word e memorizziamo la data D1 all’indirizzo 400h e la data D2 all’indirizzo 404h.

La risposta viene memorizzata all’indirizzo 408h e rappresentata come un intero nel seguente modo:

Registri usati:
R0: temporaneo per confronti

Algoritmo
Inizio

  
Caricamento dati


R0=A2;


Confronta A2, A1
                  Se (A2-A1<0) allora D2<D1;


      Se (A2-A1>0) allora D2<D1;


      Se (A1-A2=0) allora
                  Confronta M2, M2

 

Se (M2-M1<0) allora D2<D1;



Se (M2-M1>0) allora D2>D1;

                       Se (M2-M1=0) allora




Confronta G2, G1

                                         Se (G2-G1<0) allora D2<D1;



                 Se (G2-G1>0) allora D2>D1;




     Se (G2-G1=0) allora D2=D1;

Fine.
Codifica

;Programma per individuare la maggiore tra due

;date. Se D1<D2 pone in 408h il valore -1; se D1=D2 pone

;il valore 0; se D1>D2 pone il valore 1

;Autore: M.M

;Data: 20/02/2004

;File: datamagg.asm

ORG 400h


 G1 DB 10


 M1 DB 10


 A1 DW 2003


 D1 EQU 400h


 G2 DB 10


 M2 DB 10


 A2 DW 2000


 D2 EQU 404h

CODE


 MOVW A2, R0


 CMPW A1, R0   ;valuta A2-A1


 JN D2D1       ;se A2-A1<0, D2<D1


 JZ CFRMESE

         JMP D1D2 

D2D1:    MOVL #1, 408h   ;D2<D1


 JMP FINE

CFRMESE: MOVB M2, R0     ;anno uguale. 


 CMPB M1, R0     ;valuta M2-M1


 JN D2D1         ;se M2-M1<0, D2<D1 


 JZ CFRGIOR


 JMP D1D2

CFRGIOR: MOVB G2, R0     ;mese uguale


 CMPB G1, R0     ;valuta G2-G1


 JN D2D1         ;se G2-G1<0, D2<D1


 JZ UGUALI

         JMP D1D2

D1D2:  
 MOVL #-1, 408h  ;D2>D1


 JMP FINE

UGUALI:  MOVL #0, 408h

FINE:
 HALT

END

Esercizio 7
Scrivere un programma che, data una matrice quadrata MAT di dimensione NxN (N>=1) di longword memorizzata per righe a partire dalla locazione 400h, determini il numero di elementi pari

nella diagonale principale e lo memorizzi nella variabile PARI.

Soluzione

Analisi

Data una matrice NxN di LONG (4 byte) memorizzata per ;righe, l'indirizzo di MAT[i,j] è:        
Ind(MAT[i,j])=N*(i-1)*4+(j-1)*4

Quindi per N=2 Ind(MAT[2,2]=12, per N=3 Ind(MAT[2,2]=16, Ind(MAT[3,3]=32, ecc


Registri usati:

R0: contatore delle righe 
R1:temporaneo 

R2: offset (indice)

R3: temporaneo per esecuzione test

R4: contatore valori pari

Algoritmo

Inizio

    Caricamento dati

    Azzeramento registri

    Inizializzazione registri

    Ripeti

         R3=MAT[R2];

         Se R3 è pari R4=R4+1;

         R2=R2+R1   (*punta al successivo elemento *)
         R0=R0-1;

    Fino a che (R0=0);

    PARI=R4;

Fine.

Codifica

;Programma che data una matrice quadrata MAT di dimensione

;NxN (N>=1) di longword memorizzata per righe a partire 

;dalla locazione 400h, determina il numero di elementi pari

;nella diagonale principale e lo memorizza nella variabile

;PARI

;Autore: M.M.

;Data: 02/03/2004

;File:diagpari.asm

ORG 800h

MAT 
EQU 400h            ;indirizzo base della matrice

N
EQU 2               ;dimensione della matrice  

PARI
DL  0               ;variabile risultato

CODE


MOVL
#2, MAT     ;caricamento dati


MOVL
#2, MAT+4


MOVL
1, MAT+8


MOVL
#1, MAT+12

;********* azzeramento registri *********************


XORL
R0, R0      ;contatore righe


XORL
R1, R1      ;temporaneo


XORL
R2, R2      ;offset matrice


XORL
R3, R3      ;temporaneo per test pari


XORL    R4, R4      ;contatore pari

;********* inizializzazione registri ****************


MOVL
#N, R0       


MOVL
R0, R1      ;R1 <- R0=2


ADDL
#1, R1      ;R1 <- 3


ASLL
#2, R1      ;R1 <- 12

INIZIO:


MOVL
MAT(R2), R3  ;preleva valore


ANDL
#1, R3       ;test se pari


JNZ
CONT         ;se non pari


ADDL
#1, R4       ;conta valore

CONT:


ADDL
R1, R2       ;R2 <- R2 + 12


SUBL
#1, R0


JNZ
INIZIO

FINE:


MOVL
R4, PARI


HALT

END


Esercizio 8
Scrivere un programma che calcoli il quoziente Q ed il resto R della divisione intera tra due valori A e B  (supporre A,B >= 0 e A>B).
Soluzione

Analisi
La soluzione può essere calcolata eseguendo ripetutamente la sottrazione A=A-B e ogni volta incrementando di 1 un contatore (quoziente), fino a quando A>B.  In A è presente il resto R.

Registri usati:

R0: dividendo

R1: divisore

R2: quoziente

Il quoziente viene salvato in 500h, il resto in 502h.

Algoritmo
Inizio

     R0=A;

     R1=B;

     R2=0;
     Fintantoché (R0>=R1) fai

     Inizio


R0=R0-R1;

            R2=R2+1;

     Fine;

Fine.

Codifica

;Programma per il calcolo del quoziente Q e del resto R di

;una divisione intera tra due valori interi A e B.

;Il quoziente deve essere salvato in 500h, il resto in 502h

;Autore: M.M:

;Data: 26/02/2004

;File: divint.asm

ORG 400h


A DB 10    ;dividendo


B DB 3     ;divisore


Q DB 0     ;quoziente


R DB 0     ;resto

CODE


MOVB A, R0


MOVB B, R1


MOVB Q, R2

loop:
CMPB R1, R0  ;R0-R1 ?


JN fine      ;se R0<R1... 


SUBB R1, R0  ;R0=R0-R1


ADDB #1, R2  ;R2=R2+1


JNN loop

fine:
MOVB R2, 500h   ;salva quoziente


MOVB R0, 502h   ;salva resto


HALT

END

Esercizio 9
Scrivere un programma per il calcolo della media aritmetica intera di 3 valori VAR1, VAR2, VAR3.
Soluzione

Analisi

Si dichiarano 3 variabili ad esempio di tipo long e si assegnano ad esse valori iniziali.

Si sommano i 3 valori in un accumulatore e, infine, si divide la somma per 3 mediante un ciclo di sottrazioni ripetute.

Registri usati:

R0: accumulatore per la somma

R1: quantità dei numeri 

R2: contatore per il quoziente

Algoritmo
Inizio

     Caricamento valori;

     R0=somma dei valori;

     R1=3;

     R2=0;

     Ripeti

         R0=R0-R1;

         R2=R2+1;

     Fino a che (R0=0);

Fine.
;Programma per il calcolo della media aritmetica

;di 3 valori

;Autore:  M.M.
;Data: 15/01/2004

;File:media3.asm

ORG 400H


VAR1 DL 5           
;primo valore


VAR2 DL 8           
;secondo valore


VAR3 DL 2           
;terzo valore


M    DL 3              
;quantità dei valori

CODE


MOVL VAR1, R0  
;R0=VAR1


ADDL VAR2, R0       ;R0=R0+VAR2


ADDL VAR3, R0       ;R0=R0+VAR3


MOVL M, R1             ;R1=M


MOVL #0, R2             

CICLO:


SBBL R1,R0               ;ciclo di sottrazioni ripetute


ADDL #1, R2


CMPL #0, R0              ;R0=0 ?


JNZ CICLO


HALT

END

Esercizio 10
Scrivere un programma che data una sequenza di 5 byte ne imposti la parità al valore pari.

Soluzione

Analisi
Supponiamo la sequenza di byte caricata a partire dalla locazione 01Ah. Dopo caricati i valori nella sequenza, si esaminano uno ad uno e tramite operatore AND se ne verifica la parità. In caso affermativo si procede ad esaminare l’elemento successivo, altrimenti si imposta la parità a pari mediante operatore OR con il valore #80 (ottavo bit=1).
Registri usati:

R1: puntatore al singolo elemento della sequenza

R2: contatore cicli di esame degli elementi della sequenza

R3: singolo elemento della sequenza

Algoritmo

Inizio


Caricamento dati


R1=01Ah;


Fintantoché (R2<>0) fai


Inizio



R3=(R1);

            
Se R3 NON ha parità pari



Inizio







Imposta 8° bit a 1 su R3;

 



(R1)=R3      (* aggiorna il dato *)

                                    Fine



Avanza al successivo elemento

                       R2=R2-1;

             Fine; 

Codifica

;Programma che imposta la parità a pari su una

;sequenza di 5 byte caricati a partire dalla

;locazione 01Ah. 

ORG 400h

CODE


MOVL #01Ah, R1   ;R1=01ah


MOVW #4, R2

;***** caricamento dati ****************


MOVB #2h,(R1)+     ;indirizzamento con post-incremento

MOVB #5h,(R1)+


MOVB #7h,(R1)+


MOVB #9h,(R1)


MOVB #01Ah, R1

continua:


CMPB #0, R2

  
JZ fine


MOVB (R1), R3    ;R3=dato successivo


ANDB R3, R3        ;aggiorna SR


JP avanti                 ;se c'è parità procedi

        
ORB #80, R3          ;rende pari la parità 

                                            ;mediante OR tra R3 e

                                            ;1000 0000 


MOVB R3, (R1)    ;aggiorna il dato

avanti: ADDL #1, R1        ;byte successivo

        
SUBW #1, R2

        
JMP continua

fine:   
HALT

END

Esercizio 11
Dato un vettore VET di N (>=1) elementi, caricare tutte le componenti con un dato carattere

Soluzione
Analisi
Il vettore sia memorizzato a partire  dall’indirizzo 250h e sia formato da N=20 componenti di tipo byte. Lo si vuole riempire  con il carattere ‘0’ (48dec=30h).

Registri usati:

R1: offset vettore

R2: lunghezza vettore

Algoritmo
Inizio


Inizializzazione registri


Ripeti



VET[R1]=30h;



R1=R1+1;

 
            R2=R2-1;


Fino a che (R2=0);

Fine.
Codifica
;Programma che dato un vettore di 20 byte V memorizzato

;a partire dalla locazione 250h, lo riempie con tutti

;caratteri '0'

;Autore: M.M.

;Data:12/03/2004

;File:vettzero.asm

ORG 400h



DIM  EQU 14h     ;lunghezza vettore



ZERO EQU 30h     ;ascii('0')=48dec



VET  EQU 250h    ;indirizzo base vettore

CODE


;******* inizializzazione registri ************




MOVL #0h, R1     ;offset vettore



MOVB #DIM, R2

REPEAT: 


MOVB #ZERO, VET(R1)



ADDL #1, R1

UNTIL:  
SUBB #1, R2



JNZ REPEAT



HALT

END

Esercizio 12
Dati due vettori V1 e V2 di dimensione N (>=1) calcolare la loro somma e porla nel vettore V3.

Soluzione

Analisi
I vettori V1 e V2 siano di 5 elementi e memorizzati rispettivamente agli indirizzi 400h e 414h.
Il vettore V3 sia memorizzato all’indirizzo 330h.

Mediante un ciclo si calcola la somma degli elementi corrispondenti di V1 e V2 e la si pone in V3.

Registri usati:

R0: contatore elementi

R1: puntatore a V1

R2: puntatore a V2

R3: puntatore a V3

R4: temporaneo di appoggio
Algoritmo
Inizio


Inizializza registri;


Ripeti



R4=V1[R1];



R4=R4+V2[R2];



V3[R3]=R4;



R1=R1+4;



R2=R2+4;



R3=R3+4;



R0=R0-1;


Fino a che (R0=0);

Fine.

Codifica

;Programma che dati 2 vettori V1 e V2 di 5 elementi ciascuno, 

;calcoli il vettore V3 somma di V1 e V2

;Autore: M.M.

;Data:12/03/2004

;File:sommavet.asm

ORG 400h


DIM  EQU 5h      ;lunghezza vettore

        V3   EQU 330h    ;indirizzo base di V3

        V1    DL 1h, 2h, 3h, 4h, 5h


V2    DL 2h, 3h, 4h, 5h, 6h

CODE


;******* inizializzazioni ********************


MOVL #DIM, R0    ;contatore


MOVL #V1,  R1


MOVL #V2,  R2


MOVL #V3,  R3



XORL R4, R4      ;appoggio

REPEAT: MOVL (R1), R4


ADDL (R2), R4    ;somma


MOVL R4, (R3)    


ADDL #4, R1


ADDL #4, R2


ADDL #4, R3


SUBB #1, R0


JNZ REPEAT


HALT

END

Esercizio 13
Data una matrice M di word memorizzata per righe, calcolare la somma di ogni sua colonna e memorizzarla in un vettore V di longword.

Soluzione

Analisi
La matrice di word sia memorizzata all’indirizzo base 400h. Indichiamo con C ed R rispettivamente il numero di righe e colonne di M. Il vettore V abbia indirizzo base 300h. 

Mediante un ciclo si calcolano le somme di ogni colonna e le si memorizzano nelle componenti di V. La somma degli elementi di ogni colonna si calcola mediante un secondo ciclo, interno al primo, che esamina e somma tutti gli elementi appartenenti alla stessa colonna.

Registri usati:  
R0: indice matrice

R1: indice vettore

R2: contatore righe

R3: contatore colonne

R4: accumulatore per somma

R5: offset elementi colonna

R6: offset colonne

Algoritmo
Inizio


Caricamento dati matrice;


Inizializzazione registri;


Ripeti


R4=0;



R0=R6;  (*offset colonna *)



Ripeti




R4=R4+M[R0];




R0=R5;  (* riga successiva *)




R2=R2-1; 



Fino a che (R2=0);



V[R1]=R4;    (* carica valore *)



R1=R1+4;



R6=R6+2;


R3=R3-1;


Fino a che (R3=0);

Fine.
Codifica
;Programma che data una matrice di word M memorizzata 

;per righe, costruisca un vettore V di longword contenente

;in ogni componente la somma degli elementi di ciascuna

;colonna di M.

;Autore: M.M.

;Data:12/03/2004

;File:sommacol.asm

ORG 400h


R EQU 2h       ;numero righe


C EQU 3h       ;numero colonne


V EQU 300h     ;indirizzo base vettore



M DW 1,2,3,4,5,6

CODE


;******* inizializzazioni ********************


XORL R0, R0    ;indice matrice


XORL R1, R1    ;indice vettore


XORL R2, R2    ;contatore righe

           XORL R3, R3    ;contatore colonne


XORL R4, R4    ;accumulatore per somma


XORL R5, R5    ;offset elementi colonna


XORL R6, R6    ;offset colonne

           MOVL #C, R5


ASLL #1, R5    


MOVL #C, R3

CICLO1: MOVL #R, R2


XORL R4, R4


MOVL R6, R0


CICLO2:



ADDW M(R0), R4



ADDL R5, R0    ;riga successiva



SUBL #1, R2    ;decrementa numero righe



JNZ CICLO2


MOVL R4, V(R1)


ADDL #4, R1


ADDL #2, R6


SUBL #1, R3


JNZ CICLO1


HALT

END

Esercizio 14
Dato un vettore V di N elementi byte, ed un valore X, stabilire se X è presente in V, ed in caso affermativo, determinarne la posizione. In caso di insuccesso restituire il valore 0.

Soluzione

Analisi

Si carica il vettore con un dato numero di elementi. Si utilizza un ciclo che partendo dalla prima posizione, esamina le componenti una ad una. Il ciclo termina quando il vettore è stato esaminato tutto quanto (insuccesso) o quando l’elemento è stato trovato.

Registri usati:

R0: valore da cercare

R1: registro indice di V

R2: contatore dei cicli

R5: registro per il risultato

Algoritmo
Inizio


Caricamento dati


Inizializzazione registri


Fintantoché (R2<>0) e Elemento non trovato fai


Inizio



R1=R1+1;



R2=R2-1;


Fine;


Se (R2=0)

   

R5=0


Altrimenti



R5=1;

Fine.
Codifica
;Programma che effettua la ricerca di un elemento in un

;vettore di N componenti. 

;La tecnica usata è quella sequenziale.  

;Il risultato è posto in R5 e vale 0 (in caso di insuccesso)

;oppure è il valore della posizione in cui è stato trovato

;Autore: M.M.

;Data: 10/03/2004

;File:ricseq.asm

ORG 400h


DIM EQU 10


V DB 10h,25h,13h,48h,57h,61h,74h,88h,66h,9h


X DB 5h

CODE

;********* inizializzazione registri **********


MOVL #V,  R1   ;registro indice di V


MOVB X,   R0   ;valore da cercare


MOVB #DIM, R2  ;contatore cicli


MOVB #0,  R5

ciclo:  CMPB #0, R2


JZ nontrov


CMPB (R1), R0


JZ trovato


ADDL #1, R1


SUBB #1, R2


JMP ciclo

nontrov:


MOVB #0h, R5


JMP fine

trovato:ADDB #1, R1


MOVB R1, R5

fine:


HALT

END

Esercizio 15
Dato un vettore di DIM elementi (N>=1) di tipo byte memorizzato a partire dalla locazione 400h, ed un valore  N di tipo byte, determinare se N è presente in V usando il metodo di ricerca binaria.

Soluzione

Analisi
Lo spazio di ricerca iniziale è [0, DIM-1]. Il metodo consiste nel confrontare N con l’elemento centrale di V (indicato con V(centro), centro=(0+DIM-1)/2. ). Se l’elemento viene trovato in quella posizione, la ricerca termina con successo. Se N è minore di V(centro) il campo di ricerca si riduce a [0, centro-1], se N è maggiore di V(centro) il campo di ricerca si riduce a [centro+1, DIM-1].

Il ciclo di ricerca procede fino a quando o il campo di ricerca si riduce a 0 o quando N=V(centro).

Algoritmo
Inizio


Caricamento dati;


Inizializzazione registri;


Inf=0;


Sup=DIM-1;


centro=(Inf+Sup) div 2;


Fintantoché (Inf<=Sup) e (V(centro)<>N) fai


Inizio



Se (N<V(centro)) 




Sup=centro-1;



Altrimenti




Inf=centro+1;



centro=(Inf+Sup) div 2;


Fine;


Se (Inf>Sup) 


Elemento non trovato


Altrimenti



Elemento trovato in posizione centro

Fine.
Codifica
;Programma che effettua la ricerca di un elemento in un

;vettore di N componenti. 

;La tecnica usata è quella binaria

;Il risultato è posto in R5 e vale 0 (in caso di insuccesso)

;oppure è il valore della posizione in cui è stato trovato

;Autore: M.M.

;Data: 10/03/2004

;File:ricbin.asm

ORG 400h


DIM EQU 10


V DB 1,2,3,4,5,6,7,8,9,10


N DB 10

CODE

;********* inizializzazione registri **********


MOVB N,   R0    ;valore da cercare


MOVB #0, R1     ;R1 = 0  inf


MOVB #DIM-1,R2  ;R2=DIM-1   sup


MOVB #0, R5     ;risultato


MOVB R1, R6


ADDB R2, R6


ASRB #1, R6


MOVB R6, R3    ;posizione centrale

ciclo:  CMPB R1, R2


JN notrov


MOVB V(R3), R4


CMPB R0, R4


JZ trov


JN destra      ;R4<R0


JMP sinistra   ;R4>R0

destra: ADDB #1, R3


MOVB R3, R1     ;inf=centro + 1


JMP avanti

sinistra:


SUBB #1, R3


MOVB R3, R2     ;sup=centro - 1

avanti: MOVB R1, R6     ;calcolo centro


ADDB R2, R6


ASRB #1, R6


MOVB R6, R3    ;posizione centrale


JMP ciclo

trov:   MOVB R3, R5


ADDB #1, R5


JMP fine

notrov: MOVB #0, R5

fine:


HALT

END
Esercizio 16
Dato un vettore V di N elementi di tipo byte, stabilire se nel suo interno esiste un elemento pari alla somma dei rimanenti. In caso affermativo, stabilirne il valore e la posizione

Soluzione

Analisi
Il vettore viene caricato con una serie di DIM valori di tipo byte a partire dalla locazione 400h. Per individuare la presenza eventuale di un elemento di valore pari alla somma dei restanti, si può procedere nel seguente modo: si calcola in un accumulatore la somma S degli elementi di V. Successivamente si cerca se nel vettore è presente un elemento pari alla semisomma S/2. In caso affermativo si restituisce la posizione e il valore di detto elemento.

Registri usati:

R1: registro indice di V

R3: elemento corrente di V

R4: accumulatore per la somma degli elementi

Algoritmo
Inizio


Caricamento dati;

Inizializzazione registri;

Per I ( 0 a DIM-1 fai



R4=R4+V(I);


R4=R4 div 2;


R1=0;


Fintantochè (R1<DIM) e (V(R1)<>R4)



R1=R1+1;


Se (R1=DIM) 



Elemento inesistente


Altrimenti



Elemento V(R1) trovato in posizione R1;

Fine.

Codifica
;Programma che dato un vettore di N elementi memorizzati

;a partire dalla locazione 400h, determina se esiste un 

;elemento pari alla somma dei rimanenti e in caso

;affermativo ne restituisca il valore e la posizione

;rispettivamente nelle locazioni 500h e 502h

;Autore: M.M.

;Data: 12/03/2004

;File:somriman.asm

ORG 400h


V DB 1,2,3,6,0


DIM EQU 5

CODE


MOVB #0, R1     ;registro indice di V


MOVB #0, R4     ;somma elementi


MOVB #0, R3     ;elemento corrente

;*********  calcolo somma elementi **********

somma:
MOVB V(R1), R3


ADDB R3, R4


ADDB #1, R1


CMPB #DIM, R1


JNZ somma


ASRB #1, R4

;*********  ricerca elemento ****************


MOVB #0, R1

ricerca:


MOVB V(R1), R3


CMPB R3, R4


JZ esiste


ADDB #1, R1


CMPB #DIM, R1


JNZ ricerca


JZ inesist


JMP ricerca

inesist:


MOVB #0, 500h


MOVB #0, R1


JMP fine

esiste:
MOVB R3, 500h


MOVB R1, 502h

fine:


HALT

END

Esercizio 17
Calcolare la radice quadrata, approssimata all’intero per difetto, di un numero intero N (>=0).

Soluzione
Analisi
La somma dei primi N interi dispari equivale al quadrato di N. Ad esempio

N=3:  1+3+5 = 32

N=4:  1+3+5+7 = 42 

Allora, dato N, si può procedere nel seguente modo: si sottraggono ripetutamente ad N i dispari consecutivi, partendo da 1. Quando la sottrazione dà valore negativo, il numero di sottrazioni effettuate dà la radice quadrata intera di N.

Registri usati: 

R1: contatore delle sottrazioni

R2: valore dispari corrente

R3: temporaneo per la sottrazione

Algoritmo
Inizio


Assegnazione di N;


Inizializzazione registri;


R3=N;


Ripeti



Se  (R3>=R2)



Inizio




R1=R1+1;




R3=R3-R2




R2=R2+2;



Fine


Fino a che (R3<R2);

Fine.

Codifica
;Programma per il calcolo della radice quadrata di un 

;numero, approssimata all'intero per difetto

ORG 400h


N DB 5

CODE


MOVB #0, R1      ;contatore delle sottrazioni


MOVB #1, R2      ;valore dispari successivo


MOVB N, R3

ripeti:


CMPB R2, R3


JN fine


ADDB #1, R1


SUBB R2, R3


ADDB #2, R2


JMP ripeti

fine:
HALT

END


Esercizio 18
Scrivere un programma in linguaggio assembly PD32 che, data un vettore di N (>1)   elementi word operi una rotazione di 1 posizione dei suoi elementi verso l'alto. Far in modo che dopo l'esecuzione il primo elemento sia spostato in fondo al vettore.

Soluzione

Analisi 


Si carica il vettore VET a partire dalla locazione 400h con valori interi di tipo word.

Si estrae la prima componente VET[0] e la si memorizza  temporaneamente. 

Successivamente, mediante un ciclo, si avanzano le successive componenti di una posizione, fino a che l’ultima sia stata spostata al penultimo posto. A questo punto, si reinserisce la VET[0].

Registri usati:

R1: offset per elemento successivo di VET

R2: appoggio per VET[0]

R4: offset per elemento attuale di VET

R3: contatore cicli

Algoritmo

Inizio

    Azzeramento registri;

    Inizializzazione registri;

    Caricamento dati vettore;

    R2=VET[0];

    R3=N;

    I=0, J=0;

    Ripeti

       J=J+1;

      VET[J]=VET[I];

      I=I+1;

      R3=R3-1;

    Fino a che (R3=1);

    VET[I]=R2;

Fine.

Codifica

;Programma che dato un vettore VET di 5 elementi word

;memorizzato a partire dalla locazione 400h, ne opera la

;rotazione di una posizione verso l'alto, reinserendo 

;la prima componente all'ultimo posto

;Autore: M.M.

;Data: 08/03/2004

;File: rotvett.asm

ORG 800h


VET EQU 400H


N DB 5

CODE

;******** azzeramento registri ************


XORL R1, R1    ;offset per elemento successivo di VET


XORL R2, R2    ;appoggio per VET[1]


XORL R4, R4    ;offset per elemento attuale di VET


XORL R3, R3    ;contatore cicli

;******** inizializzazione registri *******


MOVB N, R3


MOVB #0, R1

;******** caricamento dati vettore ********


MOVW #1, VET


MOVW #2, VET+2


MOVW #3, VET+4


MOVW #4, VET+6


MOVW #5, VET+8


MOVW VET, R2                        ;salva primo elemento in R2


MOVW R1, R4                           ;inizializza I=J

ciclo:
ADDB #2, R1                             ;avanza R1 (J=J+1)


MOVW VET(R1), VET(R4)      ;VET[I]=VET[J]


ADDB #2, R4                              ;avanza R4 (I=I+1)


SUBB #1, R3                               ;R3=R3-1


CMPB #1, R3


JNZ ciclo                                      ;se R3<>1 ripeti 


MOVW R2, VET(R4)                  ;inserisce VET[1]


HALT

END

Esercizio 19
Scrivere un programma per calcolare la media degli elementi, di tipo word, di due vettori V1 e V2. di dimensione N (>=1). 

Soluzione
Analisi
Poiché il problema prevede di ripetere la medesima operazione su ciascuno dei due vettori, è opportuno l’utilizzo di due subroutine: una per il calcolo della somma degli elementi, una per il calcolo della media.  Si utilizza il passaggio per valore: prima della chiamata delle subroutine, si caricano i valori dei parametri di ingresso nei registri usati dalla subroutine; al termine, il registro contenente il risultato (parametro di uscita) viene salvato in un’area di memoria.
La somma degli elementi di V1 e V2 viene memorizzata nelle variabili SOMMA1 e SOMMA2; i valori delle medie vengono memorizzati nelle variabili MEDIA1 e MEDIA2.

Subroutine somma  

Input:   R3, indirizzo base del vettore

Output: R2, somma elementi del vettore

Subroutine media

Input:    R3, somma degli elementi

Output:  R1, media   

Algoritmo
Inizio


Caricamento dati;

Inizializzazione variabili

R3( #V1;

Call Somma;


SOMMA1 ( R2;


R3(R2;


Call Media;


MEDIA1( R1;


R3( #V2;


Call Somma;


SOMMA2 ( R2;


R3(R2;


Call Media;


MEDIA2( R1;

Fine.
Sub Somma
Inizio


R2 ( 0;


R1 ( 0;


Fintantoché (R1<=N)


Inizio



R2 ( R2 + (R3)+;



R1 ( R1 + 1;


Fine;



Return;
Sub Media

Inizio


R1 ( 0;


Fintantichè (R3>=N)


Inizio



R3 ( R3 – N;



R1 ( R1+1;


Fine;

Return;
Codifica

;Programma che calcola la media dei valori di due vettori

;mediante chiamata a subroutine

;Autore: M.M.

;Data:10/02/2004

;File: subrtn2.asm

Org 400h


V1 DW 2,-2,8,4       ;caricamento dati


V2 DW 5,6,-4,9


SOMMA1 DW 0     ;inizializzazione variabili


SOMMA2 DW 0


MEDIA1 DW 0


MEDIA2 DW 0


N DW 4

CODE

;****** elaborazione vettore V1 ***************


MOVL #V1, R3            ;R3 parametro input


JSR somma


MOVW R2, SOMMA1 ;R2 parametro output


MOVW R2, R3              ;R3 parametro input


JSR media


MOVW R1, MEDIA1    ;R1 parametro output

;****** elaborazione vettore V2 ***************


MOVL #V2, R3              ;R3 parametro input


JSR somma


MOVW R2, SOMMA2   ;R2 parametro output


MOVW R2, R3
     ;R3 parametro input


JSR media


MOVW R1, MEDIA2     ;R1 parametro output


JMP fine

;******************************************

;*      SUBROUTINE                                                 *

;******************************************

;
;****** somma elementi vettore ****************

;* input : R3 indirizzo base vettore                            *

;* output: R2 somma elementi                                   *

;******************************************

somma:


MOVW #0, R2                  ;R3 indirizzo vettore


MOVB #0, R1

ciclo:
CMPW N, R1


JZ finecicl


ADDW (R3)+, R2


ADDB #1, R1


JMP ciclo

finecicl: RET

;

;****** calcolo media *************************

;* input : R3 valore somma                                         *

;* output: R1 media elementi                                      *

;*******************************************

media:  MOVB #0,R1

ciclom:


CMPW N, R3


JN endm


SUBW N, R3


ADDW #1, R1


JMP ciclom

endm:   RET

fine:
HALT

END

Esercizio 20
Data una matrice quadrata di ordine N (>=2), di longword, memorizzata per righe,  calcolare il valore massimo presente nella diagonale principale e la sua posizione. 
Soluzione
Analisi
La matrice MAT si supponga caricata a partire dalla locazione 400h. Indichiamo con N la dimensione di MAT, con MAX il valore del massimo cercato e con IR e IC le coordinate della posizione di MAX.

La matrice è memorizzata per righe e quindi i valori indicati nell’esempio saranno memorizzati come sequenza: 

1, -5, 0, 12, -5, -12, -11, 7, 0, 0, 2, -1, 5, 2, 2, -5
e il programma deve dare come risultato:

MAX=2, IR=3, IC=3
Gli elementi da considerare sono quelli indicati e per essi occorre calcolare l’offset che vale (N+1)*4.
Si inizializza il massimo al primo elemento di MAT e IR=1. 

Mediante un ciclo si confrontano i restanti elementi di MAT con MAX ed eventualmente si procede all’aggiornamento.

Registri usati:

R0: temporaneo

R1: offset

R2: registro indice MAT

R3: elemento corrente di MAT

R4: massimo

R6:indice di riga/colonna

Algoritmo
Inizio


Caricamento dati;


Azzeramento registri;


Calcolo offset;


Inizializzazione dati;

Calcolo del massimo;
Fine.

Sub Calcolo Offset

Inizio

R1 ( N;


R1 ( R1+1;


R1 ( R1*4

Return;

Sub Inizializzazione dati

Inizio


R4 (MAT[1,1]


IR ( 1


R0 ( 1

Return;

Sub CalcoloMassimo

Inizio
   Ripeti


R3 ( MAT(R2);

Se (R3>R4)



Inizio




R4 ( R3;




R6 ( R0;



Fine



R2 ( R2+R1;


R0 ( R0+1;
   Fino a che (R0 > N);

   MAX ( R4;

   IC ( R6;

   IR ( R6;

Return

Codifica
;Programma che calcola il massimo della diagonale 

;principale di una matrice MAT N*N di longword ed i suoi

;indici di riga e di colonna.

ORG 400h


MAT DL 1,-5,0,12,-5,-12,-11,7,0,0,2,-1,5,2,2,-5


N EQU 4


MAX DL 0


IR DL 0


IC DL 0

CODE


JSR initregs
;azzera registri


JSR offset
;calcolo offset


JSR initdati
;inizializza dati


JSR calcmax
;calcola MAX


JMP fine

;**************************************************

;                SUBROUTINE                                                         *

;**************************************************

;******* inizializzazione registri **************** 

initregs:


XORL R0, R0
;temporaneo


XORL R1, R1
;offset


XORL R2, R2
;registro indice MAT


XORL R3, R3
;elemento corrente di MAT


XORL R4, R4
;massimo


XORL R5, R5


XORL R6, R6
;indice di riga/colonna


RET

;******* calcolo offset ***************************

offset: MOVL #N, R1


ADDL #1, R1


ASLL #2, R1


RET

;******* inizializzazione valori ******************

initdati:


MOVL MAT(R2),R4 ;max=MAT[1,1]


MOVL #1, R6     ;IR=1


MOVL #1, R0


RET

;******* calcolo MAX ******************************

calcmax:

ciclo:
MOVL MAT(R2), R3


CMPL R3, R4


JN newmax       ;R3>R4


JMP avanti

newmax:
MOVL R3, R4
;aggiorna R4 (massimo)


MOVL R0, R6
            ;aggiorna R6 (IR)

avanti:
ADDL R1, R2
                        ;successivo elemento


ADDL #1, R0


CMPL #N+1, R0


JNZ ciclo

esci:
MOVL R4, MAX


MOVL R6, IR


MOVL R6, IC

fine:
HALT

END

Esercizio 21
Data una matrice quadrata di ordine N (>=2), di longword, memorizzata per righe,  calcolare il valore minimo presente nella diagonale secondaria e la sua posizione. 

Soluzione
Analisi
La matrice MAT si supponga caricata a partire dalla locazione 400h. Indichiamo con N la dimensione di MAT, con MIN il valore del minimo cercato e con IR e IC le coordinate della posizione di MIN.

La matrice è memorizzata per righe e quindi i valori indicati nell’esempio saranno memorizzati come sequenza: 

1, -5, 0, 12, -5, -12, -11, 7, 0, 0, 2, -1, 5, 2, 2, -5
e il programma deve dare come risultato:

MIN=-11, IR=2, IC=3
Gli elementi da considerare sono quelli indicati e per essi occorre calcolare l’offset che vale (N-1)*4 .

Si inizializza il minimo all’elemento MAT(N-1)*4, IR=1. 

Mediante un ciclo si confrontano i restanti elementi di MAT con MIN ed eventualmente si procede all’aggiornamento.

Registri usati:

R0: temporaneo

R1: offset

R2: registro indice MAT

R3: elemento corrente di MAT

R4: minimo

R5: indice di colonna

R6:indice di riga

Algoritmo
Inizio


Caricamento dati;


Azzeramento registri;


Calcolo offset;


Inizializzazione dati;

Calcolo del minimo;

Fine.

Sub Calcolo Offset

Inizio


R1 ( N;


R1 ( R1-1;


R1 ( R1*4

Return;

Sub Inizializzazione dati

Inizio


R4 (MAT[1,N1]


IR ( 1


R0 ( 1

Return;

Sub CalcoloMinimo
Inizio
   Ripeti


R3 ( MAT(R2);


Se (R3<R4)



Inizio




R4 ( R3;




R6 ( R0;



Fine



R2 ( R2+R1;


R0 ( R0+1;

   Fino a che (R0 > N);

   MIN ( R4;

   IR ( R6;

   IC ( N+1-IR;

Return

Codifica
;Programma che calcola il minimo della diagonale 

;principale di una matrice MAT N*N di longword ed i suoi

;indici di riga e di colonna.

;Autore: M.M.

;Data: 15/03/2004

;File; mindiagp.asm

ORG 400h


MAT DL 1,-5,0,12,-5,-12,-11,7,0,0,2,-1,-15,2,2,-5


N EQU 4


MIN DL 0


IR DL 0


IC DL 0

CODE


JSR initregs
;azzera registri


JSR offset
;calcolo offset


JSR initdati
;inizializza dati


JSR calcmin
;calcola MIN


JMP fine

;**************************************************

;                SUBROUTINE                       *

;**************************************************

;******* inizializzazione registri **************** 

initregs:


XORL R0, R0
;temporaneo


XORL R1, R1
;offset


XORL R2, R2
;registro indice MAT


XORL R3, R3
;elemento corrente di MAT


XORL R4, R4
;minimo

XORL R5, R5 ;indice colonna


XORL R6, R6
;indice di riga


RET

;******* calcolo offset ***************************

offset: MOVL #N, R1


SUBL #1, R1


ASLL #2, R1


RET

;******* inizializzazione valori ******************

initdati:


MOVL R1, R2     ;indice primo elemento


MOVL MAT(R2),R4 ;max=MAT[1,N]


MOVL #1, R0


RET

;******* calcolo MIN ******************************

calcmin:

ciclo:
MOVL MAT(R2), R3


CMPL R4, R3


JN newmin       ;R3<R4


JMP avanti

newmin:
MOVL R3, R4
;aggiorna R4 (minimo)


MOVL R0, R6       
            ;aggiorna R6 (IR)

avanti:
ADDL R1, R2
                        ;successivo elemento


ADDL #1, R0


CMPL #N+1, R0


JNZ ciclo

esci:
MOVL R4, MIN


MOVL R6, IR


MOVL #N+1, R5   ;IC=N+1-IR


SUBL R6, R5


MOVL R5, IC   

fine:
HALT

END
Esercizio 22

Date sue stringhe S1 e S2 effettuarne il confronto e produrre il valore 0 se sono diverse, 1 se sono uguali.

Soluzione

Analisi
Per una maggiore efficienza, conviene calcolare dapprima la lunghezza di S1 e S2. Se le lunghezze sono diverse, viene restituito il valore 0. Se le lunghezze sono uguali, si confrontano uno ad uno i caratteri di S1 e S2 fino a quando sono terminate, oppure si trovino 2 caratteri diversi. Nel primo caso, le stringhe sono uguali, nel secondo sono diverse. L’esito del confronto viene posto in una variabile RISP.

Registri usati:

R0: indice stringhe

R1: lunghezza S1

R2: lunghezza S2

R3: esito: 0=diverse, 1=uguali 

R4: puntatore a stringa

R5: temporaneo

R6: temporaneo

Algoritmo
Inizio


Azzera registri


R4 ( #S2;


Call Strlen;


R2 ( R1;

R4 ( #S1;

JSR strlen


Se R1 ‹› R2


    R3 ( 0

Altrimenti


Inizio

   R0 ( R1;


   R1 ( 0;


   Fintantochè ((R0 ‹› 0) e (S1(R1) ‹› S2(R1)))


   Inizio


      R1 ( R1 + 1;


      R0 ( R0 – 1;


   Fine;


   Se (R0=0) 

R3 ( 1


   Altrimenti 



R3 ( 0;


Fine;


RISP ( R3;

Fine.

Codifica
;Programma che confronta due stringhe S1 ed S2. 

;Restituisce 0 se sono diverse, 1 se uguali

;Autore: M.M.

;Data: 15/03/2004

;File: strcmp.asm

Org 400h


S1 DB 65h,66h,67h,0h


S2 DB 65h,66h,67h,0h


RISP DB 0

CODE


JSR zeroreg


MOVL #S2, R4


JSR strlen


MOVL R1, R2


MOVL #S1, R4


JSR strlen


CMPL R1, R2


JNZ diverse


JSR cfrstr


JMP fine

diverse:MOVB #0, RISP


JMP fine

;*************************************************

;*      SUBROUTINE                               *

;*************************************************

;****** azzeramento registri *********************

zeroreg:


XORL R0, R0 ;indice stringhe


XORL R1, R1
;lunghezza S1


XORL R2, R2
;lunghezza S2


XORL R3, R3
;esito: 0=diverse, 1=uguali 


XORL R4, R4
;puntatore a stringa


XORL R5, R5
;temporaneo


XORL R6, R6
;temporaneo

;****** calcolo lunghezza S1 *********************

strlen:
MOVL #0, R1

ciclo1:
MOVL (R4), R6


CMPB #0, R6


JZ endlen1


ADDL #1, R1


ADDL #1, R4


JMP ciclo1

endlen1:


RET

;****** confronto stringhe di uguale lunghezza ***

cfrstr:

MOVL R1, R0


ADDL #1, R0


MOVL #0, R1

cont:
MOVB S1(R1), R5


MOVB S2(R1), R6


CMPB R5, R6


JNZ div


ADDB #1, R1


SUBB #1, R0


JNZ cont


JMP uguali 

div:
MOVB #0, R3


JMP finecfr

uguali: MOVB #1, R3

finecfr:RET

fine:
MOVB R3, RISP


HALT

END

Esercizio 23
Scrivere un programma in linguaggio assembly PD32 che, data una matrice MAT NxM (dove N e R sono due costanti) di longword memorizzata per colonne a partire dalla locazione 400h, ed un vettore VET di M elementi longword, memorizzato a partire dalla locazione 500h, calcoli il prodotto MAT x VET e lo memorizzi nel vettore VET2 in 600h. La matrice e il vettore possono assumere anche valori negativi. Ignorare eventuali casi di overflow.

Soluzione

Analisi
L’algoritmo segue la regola del prodotto righe*colonne. 
Registri usati:

R0: indirizzo base di MAT

R1: indirizzo base di VET

R2: indirizzo base ci VET2

R3: elemento corrente di VET

R4: accumulatore per prodotto

R5: numero righe di MAT

R6: numero colonne di MAT

Inizio


Azzeramento registri;


Caricamento dati;


Per J ( 1 a M


Inizio



I ( 1;



Ripeti




VET2(J) ( VET2(J) + MAT(J,I)*VET(J);




I ( I + 1;



Fino a che (I > N);


Fine;

Fine.

Codifica



;Data una matrice mat n*m di long-word memorizzata per 

;colonne a partire dalla locazione 800h ed un vettore vet 

;di m elementi memorizzato a partire dalla locazione 500h 

;calcolare il prodotto mat*vet e memorizzarlo in un vettore

;vet2 a partire dalla locazione 600h

;Autore:M.M.

;Data: 16/03/2004

;File:MatxVet.asm

;***********************                 ******************

ORG 800h


;Inizializzazione vettori e matrice


vet  EQU 500h


vet2 EQU 600h


mat  DL 2,3,1,3,2,5


n    EQU 3


m    EQU 2

CODE



JSR zeroregs


JSR initdata


JSR prodotto


JMP fine

;****************************************************

;*      SUBROUTINE                                  *

;****************************************************

;******* inizializzazione registri ******************

zeroregs:


XORL R0,R0


XORL R1,R1


XORL R2,R2


XORL R3,R3


XORL R4,R4


XORL R5,R5


XORL R6,R6


RET

;*******  inizializzazione dati *********************

initdata:


MOVL #mat, R0
;indirizzi base


MOVL #vet, R1


MOVL #vet2,R2


MOVL #3,(R1)+
;caricamento vet


MOVL #2,(R1)


MOVL #0,(R2)+
;azzeramento vet2


MOVL #0,(R2)+


MOVL #0,(R2)


MOVL #mat,R0
;ripristino indirizzi base



MOVL #vet,R1


MOVL #m,  R6


RET

;*******   calcolo prodotto **************************

prodotto:

colonne:MOVL #n,R5



MOVL #vet2,R2



MOVL (R1),R3


ciclo:



ADDL (R0),R4 ;prodotto per somme ripetute



SUBL #1,R3

        
JNZ ciclo


 
ADDL (R2),R4 ;incremento R4



MOVL R4,(R2) ;salva in VET2



MOVL (R1),R3 ;ripristino di r3 



ADDL #4,R0   ;avanza indice di MAT

       

ADDL #4,R2   ;avanza indice VET2



XORL R4,R4   ;azzeramento accumulatore somme



SUBL #1, R5





JNZ ciclo



ADDL #4,R1



SUBL #1,R6   ;colonna successiva


JNZ colonne

RET

fine:
HALT

END
Esercizio 24
Scrivere un programma che realizzi l’ordinamento di un vettore V di N numeri relativi secondo la tecnica del bubblesort.

Soluzione

Analisi
La tecnica del bubblesort consiste nell’ordinare un vettore mediante una serie di scambi successivi con i quali viene posto ogni volta il massimo nell’ultima posizione (indicata mediante un flag, che viene decrementato di uno ad ogni ciclo).

Ad esempio:

Man mano che si procede, i valori più grandi vanno verso la fine del vettore, fino a che, quando è rimasto soltanto un elemento, esso necessariamente si trova in prima posizione, essendo il più piccolo.

Registri usati:
R0: contatore cicli esterni

R1: indice vettore

R2: appoggio

R3: appoggio

R4: contatore cicli interni

R5: indice vettore

Algoritmo
Inizio


Caricamento dati;


Per I ( N a 2


Inizio



Flag ( I;



J ( 1;



Ripeti




Se (v(J)>v(J+1))




Inizio





Temp ( v(J);





v(J) ( v(J+1);





v(J+1) ( Temp;




Fine;




J ( J+1;



Fino a che (J = Flag);


Fine;

Fine.

;Programma per l'ordinamento in senso crescente di un vettore di N numeri relativi.

;Tecnica del bubblesort

;Autore: M.M.

;Data: 19/03/2004

;File: bubbsrt.asm

ORG 400h


v DL 5,-4,-3,12,1


n EQU 5


flag DL 0

CODE


MOVL #n, R0
;contatore cicli esterni

cicloe:
CMPL #1, R0  


JZ fine

;fine esame elementi


MOVL R0, flag


MOVL #0, R1
;indice vettore


MOVL #1, R4
;contatore cicli interni


cicloi:



MOVL R1, R5



MOVL v(R1), R2   ;R2 <- v(J)



ADDL #4, R5



MOVL v(R5), R3   ;R3 <- v(J+1)



CMPL R3, R2



JN avanti



MOVL R3, V(R1)   ;scambio



MOVL R2, V(R5)


   avanti:



ADDL #4, R1
 ;J <- J+1



ADDL #1, R4



CMPL flag,R4
 ;fine ciclo interno ?



JNZ cicloi


SUBL #1, R0


JMP cicloe

fine:


HALT

END

Esercizio 25

Scrivere un programma in assembly PD32 che, date due stringhe S1 e S2, composte da caratteri alfabetici maiuscoli e minuscoli, e terminate dal carattere ASCII 0, determini se S1<S2, S1=S2 o S1>S2, in base all’ordinamento lessicografico (alfabetico). Si consideri cha ASCII(‘A’)=6510 e che ASCII(‘a’)=9710.  Non si deve tenere conto della differenza tra maiuscole e minuscole.

Esempio

S1=”ABCD”       S2=”AbcD”    Risultato = 0   (S1=S2)
S1=”ABCDzY”   S2=“AbCD”   Risultato = 1   (S1>S2)

S1=”ABCD”       S2=”AxcD”     Risultato = -1 (S1<S2)

Soluzione
Analisi
Le due stringhe, indicate con S1 e S2, vengono costruite ponendo in ciascuna una sequenza di codici ASCII corrispondenti ai caratteri inseriti e terminando ciascuna sequenza con il carattere ASCII 0 (carattere NULLO).

Inizialmente si possono trasformare entrambe le stringhe in maiuscolo, così da rendere più semplici i calcoli successivi. Successivamente, le stringhe vengono scandite mediante un indice fino a che una delle due termini (e in questo caso essa sarà minore dell’altra) oppure si trovino due caratteri corrispondenti diversi: in questo caso, la stringa corrispondente al carattere con codice ASCII minore, sarà la stringa minore tra le due.
Registri usati:
R0: puntatore a stringa

R1: puntatore a S1

R2: puntatore a S2

R3: carattere di S1

R4: carattere di S2

R6: risposta

Algoritmo
Inizio


I ( 1;


Fintantoché (S1[I]<>0) e (S2[I]<>0) e (S1[I]<>S2[I])



I ( I + 1;


Se (S1[I]=0)



Se (S2[I]=0)    Risposta ( 0        /* stringhe uguali */



Altrimenti       Risposta ( 1         /* S1 < S2  */


Altrimenti                                                 /* S1[I]=0 */



Se (S2[I]=0)    Risposta ( -1       /* S2 < S1  */


Altrimenti                                      /* stringhe non finite */




Se (S1[I]<S2[I])  

  



  Risposta ( 1       /* S1<S2  */

                                    Altrimenti

                                                  Risposta ( -1     /* S2 < S1  */

Fine.

Codifica
;Programma che confronta due stringhe S1 ed S2 in base

;all'ordinamento lessicografico dei rispettivi caratteri

;ASCII. 

;Autore: M.M.

;Data: 19/03/2004

;File: strcmp1.asm

ORG 400h


S1 DB 65,0


S2 DB 0

CODE


JSR zeroregs

;azzera registri


JSR initdata

;inizializza puntatori


MOVL R1, R0
;parametro


JSR toupper

;toupper(S1)


MOVL R2, R0
;parametro


JSR toupper

;toupper(S2)


JSR confr

;confronto tra S1 e S2


JMP fine

;****************************************

;*      SUBROUTINE             
                    *

;****************************************

;******* azzeramento registri ***********

zeroregs:


XORL R0, R0


XORL R1, R1


XORL R2, R2


XORL R3, R3


XORL R4, R4


XORL R5, R5


RET

;******* inizializzazione dati **********

initdata:


MOVL #S1, R1


MOVL #S2, R2


RET


;******* toupper - converte in maiuscolo **

toupper:

ciclo:
MOVB (R0), R3


CMPB #0, R3


JZ endstr


CMPB #97, R3
;se R3-97>=0 il carattere




            ;è minuscolo


JN avanti
;


SUBB #32, R3
;conversione a maiuscolo


MOVB R3,(R0)          ;salva carattere

   avanti:


ADDL #1, R0


JMP ciclo

endstr: RET

;****** confronto stringhe ******************

confr:


ciclocfr:



MOVL (R1), R3
;preleva carattere



MOVL (R2), R4
;preleva carattere



CMPB #0, R3
            ;se fine S1



JZ nulS1
            ;S1=null



CMPB #0, R4
            ;S1<>null



JZ nulS2



CMPB R3, R4




JZ avanticf
             ;S1<>null e S2 <>null



JN S1S2



MOVL #-1, R6



JMP finecfr


S1S2:
MOVL #1, R6



JMP finecfr


avanticf:



ADDL #1, R1



ADDL #1, R2



JMP ciclocfr


nulS1:  CMPB #0, R4
              ;se fine S2



JZ nulS1S2
              ;S2=null



MOVL #-1, R6
  ;S1<S2



JMP finecfr


nulS1S2:



MOVL #0, R6
              ;entrambe nulle


JMP finecfr


nulS2:
MOVL #1, R6



JMP finecfr

finecfr:RET

fine:


HALT

END
Esercizio 26
Data una matrice Mat NxM, memorizzata per colonne, stabilire se esiste una riga tutta formata da zeri.

Soluzione
Analisi
Sia MAT la matrice data e sia L la dimensione dei dati (1 per byte, 2 per word, 4 per longword). L’algoritmo deve esaminare la matrice riga per riga, passando alla riga successiva ogni volta che il primo elemento di una riga risulta diverso da 0. All’inizio di ogni riga si azzera il contatore degli zeri. 

Quando una riga inizia con uno zero, si incrementa il contatore degli zeri e si passano a considerare gli elementi successivi della riga, incrementando il contatore ogni volta che si incontra uno zero. Se il contatore raggiunge il valore M (numero di elementi per riga) si esce dal ciclo e si riporta nella variabile Ind l’indice della riga individuata. Se la matrice viene scandita interamente, nella variabile Ind viene posto il valore 0.
Registri usati: 

R0: offset / indice matrice

R1: elemento corrente matrice

R3: indice di riga

R5: contatore degli zeri

Algoritmo
Inizio


R0 ( N * L;               /* calcolo offset */


Caricamento dati;


Azzeramento registri;


Verifica;
Fine.

Verifica

Inizio


R3 ( 0;        /*  indice di riga */


R5 ( 0;     /* contatore degli zeri */


Fintantochè (R3< N) e (R5<>M)


Inizio



R1 ( MAT(R0);


Se (R1<>0) 



Inizio



    R0 ( R0 + L;    /* puntatore alla riga successiva */



    R5 ( 0;



    R3 ( R3 + 1;    /* incremento indice di riga */



 Fine



Altrimenti



Inizio



    R5 ( R5 + 1;    /* conteggia lo zero */



    R0 ( R0 + offset;   /* elemento successivo della stessa riga */     



Fine;

Fine;


Se (R3=N)   Ind ( 0

Altrimenti Ind ( R3 + 1;

Fine;

Codifica
;Programma che data una matrice NxM, memorizzata

;per colonne, verifica se esiste una riga composta 

;da tutti zeri.

;Autore: M.M.

;Data: 23/03/2004

;File: rigazeri.asm

ORG 700H


Mat EQU 500H     ;dichiarazione matrice


Ind DL 0                ;indice riga formata tutta da zeri


N   DL 4                ;numero di righe


M   DL 3               ;numero delle colonne


OFF DL 0             ;offset degli elementi di riga


L   DL 4                ;dimensione del dato

CODE


JSR calcoffs


JSR loaddata


JSR zeroregs


JSR verifica


JMP fine

;*****************************************************

;*      SUBROUTINE                                                                       *

;*****************************************************

;****** calcolo offset ***********************************

calcoffs:


MOVL L,R1

ciclo:  
ADDL N,R0

           
SUBL #1,R1

        
JNZ ciclo

        
MOVL R0,OFF


RET

;****** caricamento dati nella matrice **********************
loaddata:


MOVL #MAT, R0


MOVL  #1,   (R0)+


MOVL  #4,   (R0)+  


MOVL  #7,   (R0)+


MOVL  #0,   (R0)+


MOVL  #2,   (R0)+


MOVL  #5,   (R0)+


MOVL  #8,   (R0)+


MOVL  #15, (R0)+


MOVL  #3,   (R0)+


MOVL  #6,   (R0)+


MOVL  #9,   (R0)+


MOVL  #0,   (R0)+


RET

;****** azzeramento registri *******************************
zeroregs:


XORL R0,R0


XORL R1,R1


XORL R2,R2


XORL R3,R3


XORL R4,R4


XORL R5,R5


XORL R6,R6


RET

;****** verifica dati della matrice ***************************
verifica:


MOVL #0,R3

Ciclo1: CMPL N, R3


JZ nozeri


MOVL MAT(R0),R1     ;elemento corrente

        
ADDL #0,R1          
    ;R1 è zero ?

        
JNZ  incriga                   ;avanza alla riga successiva


ADDL #1, R5                ;conteggia lo zero


CMPL M, R5


JZ exit


   ;trovata una riga di zeri
        
ADDL OFF,R0      
   ;riga successiva
        
JNZ Ciclo1


JMP endsub

exit: 
ADDL #1, R3


MOVL R3, IND            ;risultato in Ind.


JMP endsub

incriga:



ADDL L, R0


MOVL #0, R5
               ;azzera contatore di zeri


ADDL #1,R3                 ;incrementa contatore righe


JMP Ciclo1

nozeri:
MOVL #0, IND

endsub: RET

fine:
HALT

END
Esercizio 27
Calcolare la somma di due valori A e B in modo ricorsivo.

Soluzione

Analisi
Per calcolare la somma ricorsiva di due valori, occorre stabilire il passo base (ossia il problema nella sua formulazione più semplice) ed il passo induttivo (ossia la condizione in cui il problema per la sua risoluzione si serve di se stesso).

La somma può essere calcolata come segue:

Somma (A,B) = Somma (A+1, B-1)=Somma(A+2, B-2)=….=Somma(A+B,0)

quindi il passo base si verifica quando il secondo termine vale zero. In questo caso, nel primo termine avremo il valore della somma. Se il secondo termine non è zero, si richiama il calcolo della somma sul primo termine incrementato di uno e sul secondo decrementato di uno, fino al verificarsi del passo base.
Per consentire il calcolo anche con numeri relativi, occorre osservare che:
1) Se A è positivo, A va decrementato di uno e B incrementato di uno

2) Se A è negativo, A va incrementato di uno e B decrementato di uno.

Registri usati:
R1: termine A

R2: termine B

R3: somma

Algoritmo
Inizio


Caricamento dati;


Call somma;

R3 ( R2;
Fine.

Somma

Inizio


Se (R1<>0)



Inizio




Se (R1<0)

                       

Inizio





R1 ( R1 + 1;





R2 ( R2-1;




Fine




Altrimenti




Inizio





R1 ( R1 – 1;





R2 ( R2 + 1;




Fine;




Call Somma;


Altrimenti



Return;

Fine;

Codifica
;Programma per il calcolo ricorsivo della somma di due

;valori long.

;Autore: M.M.

;Data: 17/03/2004

;File:somric1.asm

Org 400h


A DL 5


B DL -3

CODE


MOVL A, R1


MOVL B, R2


JSR somma


MOVL R2, R3


JMP fine

;***** subroutine ricorsiva *************

somma:


CMPL #0, R1


JZ noric        

;passo base


ADDL #0, R1


JN R1neg


SUBL #1, R1


ADDL #1, R2


JMP ricors


R1neg: 


ADDL #1, R1


SUBL #1, R2

ricors: JSR somma

;passo induttivo

noric:
RET

;************************************

fine:
HALT

END

Esercizio 28
Data una matrice MAT  NxM, di valori interi compresi tra 0 e MAX-1, memorizzata per colonne, costruire un vettore F di MAX componenti, in modo tale che in ogni componente sia riportata la frequenza con cui il valore dell’indice di F si presenta nella matrice. 

Soluzione
Analisi
La matrice MAT è memorizzata per colonne. I valori della matrice sono interi compresi tra 0 e MAX-1. Si azzera il vettore F, formato da MAX componenti. Si calcola il numero di elementi NumEl della matrice e, tramite una scansione sequenziale, si incrementa di uno la componente di F avente indice uguale al valore corrente dell’elemento della matrice. 

Registri usati:

R0: indice matrice

R1: elemento corrente matrice

R2: temporaneo per incremento numero occorrenze

R3: contatore righe

R4, R5: calcolo indirizzo fisico nel vettore

Algoritmo
Inizio



Caricamento dati;


Azzeramento registri;


Azzeramento vettore;


Calcolo NumEl;


Costruzione F;

Fine.

Azzeramento vettore

Inizio


R0 ( 0;


Per R1 ( MAX a 1

Inizio



F(R0) ( 0;



R0 ( R0 + L;


Fine;

Fine;

Costruzione F

Inizio


R0 ( 0;


Per R3 ( NumEl a 0


Inizio



R1 ( MAT(R0);



R1 ( R1 * L;



R2 ( F(R1);



R2 ( R2 + 1;



F(R1) ( R2;



R0 ( R0 + L;


Fine;
Fine;

Codifica
;Programma che data una matrice MAT NxM, memorizzata

;per colonne, e formata da valori positivi compresi tra 

;0 e MAX-1, costruisce un vettore F con le frequenze dei 

;valori presenti in MAT 

;Autore: M.M.

;Data: 23/03/2004

;File: freqnze.asm

ORG 700H


MAT EQU 500H     ;dichiarazione matrice


F          EQU 600H


N         DL 4 
 ;numero righe


M         DL 3
 ;numero colonne


L          DL 4    ;dimensione del dato


MAX   DL 10
 ;range valori


NumEl DL 0
 ;numero elementi matrice

CODE


JSR loaddata


JSR zeroregs


JSR zerovect


JSR nelem

JSR loadvect


JMP fine

;*****************************************************

;*      SUBROUTINE                                   *

;*****************************************************

;****** caricamento dati nella matrice ***************

loaddata:


MOVL #MAT, R0


MOVL  #1, (R0)+


MOVL  #4, (R0)+  


MOVL  #7, (R0)+


MOVL  #2, (R0)+


MOVL  #4, (R0)+


MOVL  #2, (R0)+


MOVL  #1, (R0)+


MOVL  #9, (R0)+


MOVL  #4, (R0)+


MOVL  #3, (R0)+


MOVL  #3, (R0)+


MOVL  #1, (R0)


RET

;****** azzeramento registri *************************

zeroregs:


XORL R0,R0


XORL R1,R1


XORL R2,R2


XORL R3,R3


XORL R4,R4


XORL R5,R5


XORL R6,R6


RET

;****** azzeramento vettore *************************

zerovect:


MOVL #0,R0


MOVL MAX, R1

ciclo:
CMPL #0, R1


JZ endz


MOVL #0, F(R0)


ADDL L, R0


SUBL #1, R1


JMP ciclo

endz:
RET

;****** numero elementi matrice *************************

nelem:


MOVL N,R0


MOVL M,R1

Ciclo2:
CMPL #1, R1


JZ endel


ADDL N,R0


SUBL #1, R1


JMP Ciclo2

endel:
MOVL R0, NumEl

RET

;****** costruzione vettore frequenze ****************

loadvect:


MOVL #0, R0


MOVL #0,R3
;numero righe

Ciclo1: CMPL NumEl,R3


JZ exit


MOVL MAT(R0),R1     ;elemento corrente


JSR delta 


MOVL F(R1), R2


ADDL #1, R2


MOVL R2, F(R1)


ADDL L, R0


ADDL #1,R3        ;incrementa contatore righe


JMP Ciclo1

exit:
RET

;****** calcolo indirizzo fisico ****

delta:
MOVL L, R4


MOVL #0,R5

ciclo3:
CMPL #0,R4


JZ endd



ADDL R1,R5


SUBL #1,R4


JMP ciclo3

endd:
MOVL R5, R1


RET

fine:
HALT

END 
Esercizio 29
Calcolo del fattoriale di N in modo ricorsivo

Soluzione
Analisi
Il fattoriale di N può essere calcolato ricorsivamente in base alla definizione:

                                                            N! = N * (N-1)!       N>1

         




 N! =  1                      N = 1

Di conseguenza il passo base si verifica quando N=1, e in tal caso il fattoriale vale 1; se N non è uguale a uno, si inserisce N nello stack, si richiama il fattoriale sul valore  N-1 e così via, fino a raggiungere il passo base. Nello stack vengono memorizzati, oltre agli indirizzi di chiamata della subroutine, i valori che N assume via via decrescenti. Raggiunto il passo base, che calcola il fattoriale di 1, occorre estrarre dallo stack l’ultimo valore e moltiplicarlo per il fattoriale calcolato e così via fino a concludere la prima chiamata della subroutine. Il fattoriale viene posto nel registro R1.

Registri usati:

R0: valore di N

R1: valore del fattoriale

R2: temporaneo per la moltiplicazione

Algoritmo
Inizio


R0 ( N;


Call fattr

Fine.

Fattr

Inizio


Se (R0=1)



R1 ( 1

Altrimenti


Inizio



Push R0;



R0 ( R0 - 1;



Call Fattr;



Pop R0;



R1 ( R0 * R1;


Fine


Return

Fine;

Codifica

ORG 400h


N DL 5

CODE


MOVL N, R0


JSR fattr


JMP fine

;****** subroutine ricorsiva per il fattoriale ****

fattr:
CMPL #1, R0


JZ pbase


;passo base


PUSH R0


SUBL #1, R0


JSR fattr


;passo induttivo


POP R0


MOVL #0, R2




mult:
CMPL #0, R0

;R1=R0*R1



JZ endmult



ADDL R1, R2



SUBL #1, R0



JMP mult


endmult:MOVL R2, R1



JMP endf



pbase:
MOVL #1, R1

;fatt(1)=1



JMP endf

endf:
RET

fine:
HALT

END

